
// PRŮVODCE 3 - Zadání základních geometrických útvarů
    // V této části půjde o seznámení se zadávání bodů, 
        // přímek, polopřímek a úseček. 

hlavni () {
    zobrazeni("mongeovo");
    papir(29.7, 21, 1); //A4
    okraj(0.75,0.75,0.75,0.75,0.25,1,0);
    posunutipss(0,0);    
    nadpis("", "", datum(), "", "");

    // Bod může zadat takto:
    A = bod(1, 3, 4);        // 
    B = bod(2, 5);        // Tento bod leží v půdorysně.
    
    S = (A + B) / 2;        // Takto získáme střed.
    
    // Je možné zjistit souřadnice bodu a pak s nimi pracovat.
    i = S."x";        
      
    a = primka(A, B);        // Přímka určená body A a B.
    
    C = bod(4, 3, 1);        // Další body.
    D = bod(-5, 1, 4);        
    
    b = usecka(C, D);
    
    R = p(a, b); // Průnikem je nic, jsou to mimoběžky.
        // Pokud chceme bod, ve kterém se protínají v půdorysu, 
        // musíme použít funkci p s půdorysy přímek:
    P = p(a.1, b.1); // Průnikem je bod.
    c = primka(a, b); // Tato přímka má půdorys jako a, nárys jako b.

        // Kolmice v půdorysně k a (půdorysem) bodem B (půdorysem).
    k = kolmice(Pi, a.1, B.1); 
        // Přímky procházející B a svírající s a 80 stupňů v půdorysně.
    l = primka(Pi, a.1, B.1, 80); 
        
        // Polopřímku zadaáme takto:
    a = poloprimka(D, A);
        
        // U přímek (také polopřímek a úceček) můžeme zjistit jejich zadávací
        // body, vektor, půdorys, nárys a bokorys.
        // a."A", a."B", a."v", a.1, a.2, a.3
}


